
1 Introduction

ParenScript is a simple language that looks a lot like Lisp, but actually is JavaScript
in disguise. Actually, it is JavaScript embedded in a host Lisp. This way,
JavaScript programs can be seamlessly integrated in a Lisp web application.
The programmer doesn’t have to resort to a different syntax, and JavaScript
code can easily be generated without having to resort to complicated string
generation or FORMAT expressions.

An example is worth more than a thousand words. The following Lisp
expression is a call to the ParenScript “compiler”. The ParenScript “compiler”
transforms the expression in ParenScript into an equivalent, human-readable
expression in JavaScript.

(ps

(defun foobar (a b)

(return (+ a b))))

The resulting javascript is:

"

function foobar(a, b) {

return a + b;

}

"

Great care has been given to the indentation and overall readability of the gen-
erated JavaScript code.

2 Features

ParenScript supports all the statements and expressions defined by the Ec-
maScript 262 standard. Lisp symbols are converted to camelcase, javascript-
compliant syntax. This idea is taken from Linj by Antonio Menezes Leitao.
Here are a few examples of Lisp symbol to JavaScript name conversion:

(js-to-string ’foobar) => "foobar"

(js-to-string ’foo-bar) => "fooBar"

(js-to-string ’foo-b-@-r) => "fooBAtR"

(js-to-string ’foo-b@r) => "fooBatr"

(js-to-string ’*array) => "Array"

(js-to-string ’*math.floor) => "Math.floor"

It also supports additional iteration constructs, relieving the programmer of the
burden of iterating over arrays. for loops can be written using the customary
DO syntax.

(ps

(do ((i 0 (incf i))

(j (aref arr i) (aref arr i)))

((>= i 10))

(alert (+ "i is " i " and j is " j))))

; compiles to

"

for (var i = 0, j = arr[i]; i < 10; i = ++i, j = arr[i]) {

alert(’i is ’ + i + ’ and j is ’ + j);

}

"

ParenScript uses the Lisp reader, allowing for reader macros. It also comes
with its own macro environment, allowing host Lisp macros and ParenScript
macros to coexist without interfering with each other. For example, the 1+
construct is implemented using a ParenScript macro:

(defpsmacro 1+ (form)

‘(+ ,form 1))

ParenScript allows the creation of JavaScript objects in a Lispy way, using key-
word arguments.

(ps

(create :foo "foo"

:bla "bla"))

; compiles to

"

{ foo : ’foo’,

bla : ’bla’ }

"

ParenScript features a HTML generator. Using the same syntax as the HTML-
GEN package of Franz, Inc., it can generate JavaScript string expressions. This
allows for a clean integration of HTML in ParenScript code, instead of writing
the tedious and error-prone string generation code generally found in JavaScript.

(ps

(defun add-div (name href link-text)

(document.write

(ps-html ((:div :id name)

"The link is: "

((:a :href href) link-text))))))

; compiles to

"

function addDiv(name, href, linkText) {

document.write(’<div id=\"’ + name + ’\">The link is: <a href=\"’

+ href + ’\">’

+ linkText + ’</a></div>’);

}

"

In order to have a complete web application framework available in Lisp, Paren-
Script also provides a sexp-based syntax for CSS stylesheets. Thus, a complete
web application featuring HTML, CSS and JavaScript documents can be gener-
ated using Lisp syntax, allowing the programmer to use Lisp macros to factor
out the redundancies and complexities of Web syntax. For example, to gener-
ate a CSS inline node in a HTML document using the AllegroServe HTMLGEN
library:

(html-stream *standard-output*

(html

(:html

(:head

(css (* :border "1px solid black")

(div.bl0rg :font-family "serif")

(("a:active" "a:hoover") :color "black" :size "200%"))))))

; which produces

<html><head><style type="text/css">

<!--

* {

border:1px solid black;

}

div.bl0rg {

font-family:serif;

}

a:active,a:hoover {

color:black;

size:200%;

}

-->

</style>

</head>

</html>

3 ParenScript Tutorial

This chapter is a short introductory tutorial to ParenScript. It hopefully will
give you an idea how ParenScript can be used in a Lisp web application.

4 Setting up the ParenScript environment

In this tutorial, we will use the Portable Allegroserve webserver to serve the
tutorial web application. We use the ASDF system to load both Allegroserve
and ParenScript. I assume you have installed and downloaded Allegroserve
and Parenscript, and know how to setup the central registry for ASDF.

(asdf:oos ’asdf:load-op :aserve)

; ... lots of compiler output ...

(asdf:oos ’asdf:load-op :parenscript)

; ... lots of compiler output ...

The tutorial will be placed in its own package, which we first have to define.

(defpackage :js-tutorial

(:use :common-lisp :net.aserve :net.html.generator :parenscript))

(in-package :js-tutorial)

The next command starts the webserver on the port 8080.

(start :port 8080)

We are now ready to generate the first JavaScript-enabled webpages using
ParenScript.

5 A simple embedded example

The first document we will generate is a simple HTML document, which fea-
tures a single hyperlink. When clicking the hyperlink, a JavaScript handler

opens a popup alert window with the string “Hello world”. To facilitate the
development, we will factor out the HTML generation to a separate function,
and setup a handler for the url “/tutorial1”, which will generate HTTP headers
and call the function TUTORIAL1. At first, our function does nothing.

(defun tutorial1 (req ent)

(declare (ignore req ent))

nil)

(publish :path "/tutorial1"

:content-type "text/html; charset=ISO-8859-1"

:function (lambda (req ent)

(with-http-response (req ent)

(with-http-body (req ent)

(tutorial1 req ent)))))

Browsing “http://localhost:8080/tutorial1” should return an empty HTML page.
It’s now time to fill this rather page with content. ParenScript features a macro
that generates a string that can be used as an attribute value of HTML nodes.

(defun tutorial1 (req ent)

(declare (ignore req ent))

(html

(:html

(:head (:title "ParenScript tutorial: 1st example"))

(:body (:h1 "ParenScript tutorial: 1st example")

(:p "Please click the link below." :br

((:a :href "#" :onclick (ps-inline

(alert "Hello World")))

"Hello World"))))))

Browsing “http://localhost:8080/tutorial1” should return the following HTML:

<html><head><title>ParenScript tutorial: 1st example</title>

</head>

<body><h1>ParenScript tutorial: 1st example</h1>

<p>Please click the link below.<br/>

<a href="#"

onclick="javascript:alert(&quot;Hello World&quot;);">Hello World</a>

</p>

</body>

</html>

6 Adding an inline ParenScript

Suppose we now want to have a general greeting function. One way to do
this is to add the javascript in a SCRIPT element at the top of the HTML page.
This is done using the JS-SCRIPT macro (defined below) which will generate
the necessary XML and comment tricks to cleanly embed JavaScript. We will
redefine our TUTORIAL1 function and add a few links:

(defmacro js-script (&rest body)

"Utility macro for including ParenScript into the HTML notation

of net.html.generator library that comes with AllegroServe."

‘((:script :type "text/javascript")

(:princ (format nil "~%// <![CDATA[~%"))

(:princ (ps ,@body))

(:princ (format nil "~%// ]]>~%"))))

(defun tutorial1 (req ent)

(declare (ignore req ent))

(html

(:html

(:head

(:title "ParenScript tutorial: 2nd example")

(js-script

(defun greeting-callback ()

(alert "Hello World"))))

(:body

(:h1 "ParenScript tutorial: 2nd example")

(:p "Please click the link below." :br

((:a :href "#" :onclick (ps-inline (greeting-callback)))

"Hello World")

:br "And maybe this link too." :br

((:a :href "#" :onclick (ps-inline (greeting-callback)))

"Knock knock")

:br "And finally a third link." :br

((:a :href "#" :onclick (ps-inline (greeting-callback)))

"Hello there"))))))

This will generate the following HTML page, with the embedded JavaScript
nicely sitting on top. Take note how GREETING-CALLBACK was converted to
camelcase, and how the lispy DEFUN was converted to a JavaScript function
declaration.

<html><head><title>ParenScript tutorial: 2nd example</title>

<script type="text/javascript">

// <![CDATA[

function greetingCallback() {

alert("Hello World");

}

// ]]>

</script>

</head>

<body><h1>ParenScript tutorial: 2nd example</h1>

<p>Please click the link below.<br/>

<a href="#"

onclick="javascript:greetingCallback();">Hello World</a>

<br/>

And maybe this link too.<br/>

<a href="#"

onclick="javascript:greetingCallback();">Knock knock</a>

<br/>

And finally a third link.<br/>

<a href="#"

onclick="javascript:greetingCallback();">Hello there</a>

</p>

</body>

</html>

7 Generating a JavaScript file

The best way to integrate ParenScript into a Lisp application is to generate a
JavaScript file from ParenScript code. This file can be cached by intermediate
proxies, and webbrowsers won’t have to reload the JavaScript code on each
pageview. We will publish the tutorial JavaScript under “/tutorial.js”.

(defun tutorial1-file (req ent)

(declare (ignore req ent))

(html (:princ

(ps (defun greeting-callback ()

(alert "Hello World"))))))

(publish :path "/tutorial1.js"

:content-type "text/javascript; charset=ISO-8859-1"

:function (lambda (req ent)

(with-http-response (req ent)

(with-http-body (req ent)

(tutorial1-file req ent)))))

(defun tutorial1 (req ent)

(declare (ignore req ent))

(html

(:html

(:head

(:title "ParenScript tutorial: 3rd example")

((:script :language "JavaScript" :src "/tutorial1.js")))

(:body

(:h1 "ParenScript tutorial: 3rd example")

(:p "Please click the link below." :br

((:a :href "#" :onclick (ps-inline (greeting-callback)))

"Hello World")

:br "And maybe this link too." :br

((:a :href "#" :onclick (ps-inline (greeting-callback)))

"Knock knock")

:br "And finally a third link." :br

((:a :href "#" :onclick (ps-inline (greeting-callback)))

"Hello there"))))))

This will generate the following JavaScript code under “/tutorial1.js”:

function greetingCallback() {

alert("Hello World");

}

and the following HTML code:

<html><head><title>ParenScript tutorial: 3rd example</title>

<script language="JavaScript" src="/tutorial1.js"></script>

</head>

<body><h1>ParenScript tutorial: 3rd example</h1>

<p>Please click the link below.<br/>

<a href="#" onclick="javascript:greetingCallback();">Hello World</a>

<br/>

And maybe this link too.<br/>

<a href="#" onclick="javascript:greetingCallback();">Knock knock</a>

<br/>

And finally a third link.<br/>

<a href="#" onclick="javascript:greetingCallback();">Hello there</a>

</p>

</body>

</html>

8 A ParenScript slideshow

While developing ParenScript, I used JavaScript programs from the web and
rewrote them using ParenScript. This is a nice slideshow example from

http://www.dynamicdrive.com/dynamicindex14/dhtmlslide.htm

The slideshow will be accessible under “/slideshow”, and will slide through
the images “photo1.png”, “photo2.png” and “photo3.png”. The first Paren-
Script version will be very similar to the original JavaScript code. The second
version will then show how to integrate data from the Lisp environment into
the ParenScript code, allowing us to customize the slideshow application by
supplying a list of image names. We first setup the slideshow path.

(publish :path "/slideshow"

:content-type "text/html"

:function (lambda (req ent)

(with-http-response (req ent)

(with-http-body (req ent)

(slideshow req ent)))))

(publish :path "/slideshow.js"

:content-type "text/html"

:function (lambda (req ent)

(with-http-response (req ent)

(with-http-body (req ent)

(js-slideshow req ent)))))

The images are just random files I found on my harddrive. We will publish
them by hand for now.

(publish-file :path "/photo1.jpg"

:file "/home/viper/photo1.jpg")

(publish-file :path "/photo2.jpg"

:file "/home/viper/photo2.jpg")

(publish-file :path "/photo3.jpg"

:file "/home/viper/photo3.jpg")

The function SLIDESHOW generates the HTML code for the main slideshow
page. It also features little bits of ParenScript. These are the callbacks on the
links for the slideshow application. In this special case, the javascript generates
the links itself by using document.write in a “SCRIPT” element. Users that
don’t have JavaScript enabled won’t see anything at all.

SLIDESHOW also generates a static array called PHOTOS which holds the
links to the photos of the slideshow. This array is handled by the ParenScript
code in “slideshow.js”. Note how the HTML code issued by ParenScrip is gen-
erated using the PS-HTML construct. In fact, there are two different HTML
generators in the example below, one is the AllegroServe HTML generator, and
the other is the ParenScript standard library HTML generator, which produces
a JavaScript expression which evaluates to an HTML string.

(defun slideshow (req ent)

(declare (ignore req ent))

(html

(:html

(:head (:title "ParenScript slideshow")

((:script :language "JavaScript"

:src "/slideshow.js"))

(js-script

(defvar *linkornot* 0)

(defvar photos (array "photo1.jpg"

"photo2.jpg"

"photo3.jpg"))))

(:body (:h1 "ParenScript slideshow")

(:body (:h2 "Hello")

((:table :border 0

:cellspacing 0

:cellpadding 0)

(:tr ((:td :width "100%" :colspan 2 :height 22)

(:center

(js-script

(let ((img (ps-html

((:img :src (aref photos 0)

:name "photoslider"

:style (+ "filter:"

(lisp (ps (reveal-trans

(setf duration 2)

(setf transition 23)))))

:border 0)))))

(document.write

(if (= *linkornot* 1)

(ps-html ((:a :href "#"

:onclick (lisp (ps-inline (transport))))

img))

img)))))))

(:tr ((:td :width "50%" :height "21")

((:p :align "left")

((:a :href "#"

:onclick (ps-inline (backward)

(return false)))

"Previous Slide")))

((:td :width "50%" :height "21")

((:p :align "right")

((:a :href "#"

:onclick (ps-inline (forward)

(return false)))

"Next Slide"))))))))))

SLIDESHOW generates the following HTML code (long lines have been bro-
ken):

<html><head><title>ParenScript slideshow</title>

<script language="JavaScript" src="/slideshow.js"></script>

<script type="text/javascript">

// <![CDATA[

var LINKORNOT = 0;

var photos = [ "photo1.jpg", "photo2.jpg", "photo3.jpg" ];

// ]]>

</script>

</head>

<body><h1>ParenScript slideshow</h1>

<body><h2>Hello</h2>

<table border="0" cellspacing="0" cellpadding="0">

<tr><td width="100%" colspan="2" height="22">

<center><script type="text/javascript">

// <![CDATA[

var img =

"<img src=\"" + photos[0]

+ "\" name=\"photoslider\"

style=\"filter:revealTrans(duration=2,transition=23)\"

border=\"0\"></img>";

document.write(LINKORNOT == 1 ?

"<a href=\"#\"

onclick=\"javascript:transport()\">"

+ img + "</a>"

: img);

// ]]>

</script>

</center>

</td>

</tr>

<tr><td width="50%" height="21"><p align="left">

<a href="#"

onclick="javascript:backward(); return false;">Previous Slide</a>

</p>

</td>

<td width="50%" height="21"><p align="right">

<a href="#"

onclick="javascript:forward(); return false;">Next Slide</a>

</p>

</td>

</tr>

</table>

</body>

</body>

</html>

The actual slideshow application is generated by the function JS-SLIDESHOW,
which generates a ParenScript file. Symbols are converted to JavaScript vari-
ables, but the dot “.” is left as is. This enables convenient access to object slots
without using the SLOT-VALUE function all the time. However, when the ob-
ject we are referring to is not a variable, but for example an element of an array,
we have to revert to SLOT-VALUE.

(defun js-slideshow (req ent)

(declare (ignore req ent))

(html

(:princ

(ps

(defvar *preloaded-images* (make-array))

(defun preload-images (photos)

(dotimes (i photos.length)

(setf (aref *preloaded-images* i) (new *Image)

(slot-value (aref *preloaded-images* i) ’src)

(aref photos i))))

(defun apply-effect ()

(when (and document.all photoslider.filters)

(let ((trans photoslider.filters.reveal-trans))

(setf (slot-value trans ’*Transition)

(floor (* (random) 23)))

(trans.stop)

(trans.apply))))

(defun play-effect ()

(when (and document.all photoslider.filters)

(photoslider.filters.reveal-trans.play)))

(defvar *which* 0)

(defun keep-track ()

(setf window.status

(+ "Image " (1+ *which*) " of " photos.length)))

(defun backward ()

(when (> *which* 0)

(decf *which*)

(apply-effect)

(setf document.images.photoslider.src

(aref photos *which*))

(play-effect)

(keep-track)))

(defun forward ()

(when (< *which* (1- photos.length))

(incf *which*)

(apply-effect)

(setf document.images.photoslider.src

(aref photos *which*))

(play-effect)

(keep-track)))

(defun transport ()

(setf window.location (aref photoslink *which*)))))))

JS-SLIDESHOW generates the following JavaScript code:

var PRELOADEDIMAGES = new Array();

function preloadImages(photos) {

for (var i = 0; i != photos.length; i = i++) {

PRELOADEDIMAGES[i] = new Image;

PRELOADEDIMAGES[i].src = photos[i];

}

}

function applyEffect() {

if (document.all && photoslider.filters) {

var trans = photoslider.filters.revealTrans;

trans.Transition = Math.floor(Math.random() * 23);

trans.stop();

trans.apply();

}

}

function playEffect() {

if (document.all && photoslider.filters) {

photoslider.filters.revealTrans.play();

}

}

var WHICH = 0;

function keepTrack() {

window.status = "Image " + (WHICH + 1) + " of " +

photos.length;

}

function backward() {

if (WHICH > 0) {

--WHICH;

applyEffect();

document.images.photoslider.src = photos[WHICH];

playEffect();

keepTrack();

}

}

function forward() {

if (WHICH < photos.length - 1) {

++WHICH;

applyEffect();

document.images.photoslider.src = photos[WHICH];

playEffect();

keepTrack();

}

}

function transport() {

window.location = photoslink[WHICH];

}

9 Customizing the slideshow

For now, the slideshow has the path to all the slideshow images hardcoded in
the HTML code, as well as in the publish statements. We now want to cus-
tomize this by publishing a slideshow under a certain path, and giving it a list
of image urls and pathnames where those images can be found. For this, we
will create a function PUBLISH-SLIDESHOW which takes a prefix as argument,
as well as a list of image pathnames to be published.

(defun publish-slideshow (prefix images)

(let* ((js-url (format nil "~Aslideshow.js" prefix))

(html-url (format nil "~Aslideshow" prefix))

(image-urls

(mapcar (lambda (image)

(format nil "~A~A.~A" prefix

(pathname-name image)

(pathname-type image)))

images)))

(publish :path html-url

:content-type "text/html"

:function (lambda (req ent)

(with-http-response (req ent)

(with-http-body (req ent)

(slideshow2 req ent image-urls)))))

(publish :path js-url

:content-type "text/html"

:function (lambda (req ent)

(with-http-response (req ent)

(with-http-body (req ent)

(js-slideshow req ent)))))

(map nil (lambda (image url)

(publish-file :path url

:file image))

images image-urls)))

(defun slideshow2 (req ent image-urls)

(declare (ignore req ent))

(html

(:html

(:head (:title "ParenScript slideshow")

((:script :language "JavaScript"

:src "/slideshow.js"))

((:script :type "text/javascript")

(:princ (format nil "~%// <![CDATA[~%"))

(:princ (ps (defvar *linkornot* 0)))

(:princ (ps* ‘(defvar photos (array ,@image-urls))))

(:princ (format nil "~%// ]]>~%"))))

(:body (:h1 "ParenScript slideshow")

(:body (:h2 "Hello")

((:table :border 0

:cellspacing 0

:cellpadding 0)

(:tr ((:td :width "100%" :colspan 2 :height 22)

(:center

(js-script

(let ((img (ps-html

((:img :src (aref photos 0)

:name "photoslider"

:style (+ "filter:"

(lisp (ps (reveal-trans

(setf duration 2)

(setf transition 23)))))

:border 0)))))

(document.write

(if (= *linkornot* 1)

(ps-html ((:a :href "#"

:onclick (lisp (ps-inline (transport))))

img))

img)))))))

(:tr ((:td :width "50%" :height "21")

((:p :align "left")

((:a :href "#"

:onclick (ps-inline (backward)

(return false)))

"Previous Slide")))

((:td :width "50%" :height "21")

((:p :align "right")

((:a :href "#"

:onclick (ps-inline (forward)

(return false)))

"Next Slide"))))))))))

We can now publish the same slideshow as before, under the “/bknr/” prefix:

(publish-slideshow "/bknr/"

‘("/home/viper/photo1.jpg" "/home/viper/photo2.jpg" "/home/viper/photo3.jpg"))

That’s it, we can now access our customized slideshow under

http://localhost:8080/bknr/slideshow

10 ParenScript Language Reference

This chapters describes the core constructs of ParenScript, as well as its com-
pilation model. This chapter is aimed to be a comprehensive reference for
ParenScript developers. Programmers looking for how to tweak the Paren-
Script compiler itself should turn to the ParenScript Internals chapter.

11 Statements and Expressions

In contrast to Lisp, where everything is an expression, JavaScript makes the
difference between an expression, which evaluates to a value, and a statement,
which has no value. Examples for JavaScript statements are for, with and
while. Most ParenScript forms are expression, but certain special forms are
not (the forms which are transformed to a JavaScript statement). All Paren-
Script expressions are statements though. Certain forms, like IF and PROGN,

generate different JavaScript constructs whether they are used in an expression
context or a statement context. For example:

(+ i (if 1 2 3)) => i + (1 ? 2 : 3)

(if 1 2 3)

=> if (1) {

2;

} else {

3;

}

12 Symbol conversion

Lisp symbols are converted to JavaScript symbols by following a few simple
rules. Special characters !, ?, #, @, %, ’/’, * and + get replaced by their written-
out equivalents “bang”, “what”, “hash”, “at”, “percent”, “slash”, “start” and
“plus” respectively. The $ character is untouched.

!?#@% => bangwhathashatpercent

The - is an indication that the following character should be converted to up-
percase. Thus, - separated symbols are converted to camelcase. The character
however is left untouched.

bla-foo-bar => blaFooBar

If you want a JavaScript symbol beginning with an uppercase, you can either
use a leading -, which can be misleading in a mathematical context, or a lead-
ing *.

*array => Array

The . character is left as is in symbols. This allows the ParenScript program-
mer to use a practical shortcut when accessing slots or methods of JavaScript
objects. Instead of writing

(slot-value foobar ’slot)

we can write

foobar.slot

A symbol beggining and ending with + or * is converted to all uppercase, to
signify that this is a constant or a global variable.

*global-array* => GLOBALARRAY

*global-array*.length => GLOBALARRAY.length

16



12.1 Reserved Keywords

The following keywords and symbols are reserved in ParenScript, and should
not be used as variable names.

! ~ ++ -- * / % + - << >> >>> < > <= >= == != ==== !== & ^ | && ||

*= /= %= += -= <<= >>= >>>= &= ^= |= 1- 1+

ABSTRACT AND AREF ARRAY BOOLEAN BREAK BYTE CASE CATCH CC-IF CHAR CLASS

COMMA CONST CONTINUE CREATE DEBUGGER DECF DEFAULT DEFUN DEFVAR DELETE

DO DOEACH DOLIST DOTIMES DOUBLE ELSE ENUM EQL EXPORT EXTENDS FALSE

FINAL FINALLY FLOAT FLOOR FOR FUNCTION GOTO IF IMPLEMENTS IMPORT IN INCF

INSTANCEOF INT INTERFACE JS LAMBDA LET LISP LIST LONG MAKE-ARRAY NATIVE NEW

NIL NOT OR PACKAGE PRIVATE PROGN PROTECTED PUBLIC RANDOM REGEX RETURN

SETF SHORT SLOT-VALUE STATIC SUPER SWITCH SYMBOL-MACROLET SYNCHRONIZED T

THIS THROW THROWS TRANSIENT TRY TYPEOF UNDEFINED UNLESS VAR VOID VOLATILE

WHEN WHILE WITH WITH-SLOTS

13 Literal values

13.1 Number literals
; number ::= a Lisp number

ParenScript supports the standard JavaScript literal values. Numbers are
compiled into JavaScript numbers.

1 => 1

123.123 => 123.123

Note that the base is not conserved between Lisp and JavaScript.

#x10 => 16

13.2 String literals
; string ::= a Lisp string

Lisp strings are converted into JavaScript literals.

"foobar" => ’foobar’

"bratzel bub" => ’bratzel bub’

Escapes in Lisp are not converted to JavaScript escapes. However, to avoid
having to use double backslashes when constructing a string, you can use the
CL-INTERPOL library by Edi Weitz.

13.3 Array literals
; (ARRAY {values}*)

; (MAKE-ARRAY {values}*)

; (AREF array index)

;

; values ::= a ParenScript expression

; array ::= a ParenScript expression

; index ::= a ParenScript expression

Array literals can be created using the ARRAY form.

(array) => [ ]

(array 1 2 3) => [ 1, 2, 3 ]

(array (array 2 3)

(array "foobar" "bratzel bub"))

=> [ [ 2, 3 ], [ ’foobar’, ’bratzel bub’ ] ]

Arrays can also be created with a call to the Array function using the MAKE-ARRAY.
The two forms have the exact same semantic on the JavaScript side.

(make-array) => new Array()

(make-array 1 2 3) => new Array(1, 2, 3)

(make-array

(make-array 2 3)

(make-array "foobar" "bratzel bub"))

=> new Array(new Array(2, 3), new Array(’foobar’, ’bratzel bub’))

Indexing arrays in ParenScript is done using the form AREF. Note that JavaScript
knows of no such thing as an array. Subscripting an array is in fact reading a
property from an object. So in a semantic sense, there is no real difference be-
tween AREF and SLOT-VALUE.

13.4 Object literals
; (CREATE {name value}*)

; (SLOT-VALUE object slot-name)

; (WITH-SLOTS ({slot-name}*) object body)

;

; name ::= a ParenScript symbol or a Lisp keyword

; value ::= a ParenScript expression

; object ::= a ParenScript object expression

; slot-name ::= a quoted Lisp symbol

; body ::= a list of ParenScript statements

Object literals can be create using the CREATE form. Arguments to the
CREATE form is a list of property names and values. To be more “lispy”, the
property names can be keywords.

(create :foo "bar" :blorg 1)

=> { foo : ’bar’,

blorg : 1 }

(create :foo "hihi"

:blorg (array 1 2 3)

:another-object (create :schtrunz 1))

=> { foo : ’hihi’,

blorg : [ 1, 2, 3 ],

anotherObject : { schtrunz : 1 } }

Object properties can be accessed using the SLOT-VALUE form, which takes an
object and a slot-name.

(slot-value an-object ’foo) => anObject.foo

A programmer can also use the “.” symbol notation explained above.

an-object.foo => anObject.foo

The form WITH-SLOTS can be used to bind the given slot-name symbols to a
macro that will expand into a SLOT-VALUE form at expansion time.

(with-slots (a b c) this

(+ a b c))

=> this.a + this.b + this.c;

13.5 Regular Expression literals
; (REGEX regex)

;

; regex ::= a Lisp string

Regular expressions can be created by using the REGEX form. If the argument
does not start with a slash, it is surrounded by slashes to make it a proper
JavaScript regex. If the argument starts with a slash it is left as it is. This makes
it possible to use modifiers such as slash-i (case-insensitive) or slash-g (match-
globally (all)).

(regex "foobar") => /foobar/

(regex "/foobar/i") => /foobar/i

Here CL-INTERPOL proves really useful.

(regex #?r"/([^\s]+)foobar/i") => /([^\s]+)foobar/i

13.6 Literal symbols
; T, FALSE, NIL, UNDEFINED, THIS

The Lisp symbols T and FALSE are converted to their JavaScript boolean equiv-
alents true and false.

T => true

FALSE => false

The Lisp symbol NIL is converted to the JavaScript keyword null.

NIL => null

The Lisp symbol UNDEFINED is converted to the JavaScript keyword undefined.

UNDEFINED => undefined

The Lisp symbol THIS is converted to the JavaScript keyword this.

THIS => this

14 Variables
; variable ::= a Lisp symbol

All the other literal Lisp values that are not recognized as special forms or
symbol macros are converted to JavaScript variables. This extreme freedom is
actually quite useful, as it allows the ParenScript programmer to be flexible, as
flexible as JavaScript itself.

variable => variable

a-variable => aVariable

*math => Math

*math.floor => Math.floor

15 Function calls and method calls
; (function {argument}*)

; (method object {argument}*)

;

; function ::= a ParenScript expression or a Lisp symbol

; method ::= a Lisp symbol beginning with .

; object ::= a ParenScript expression

; argument ::= a ParenScript expression

Any list passed to the JavaScript that is not recognized as a macro or a spe-
cial form (see “Macro Expansion” below) is interpreted as a function call. The
function call is converted to the normal JavaScript function call representation,
with the arguments given in paren after the function name.

(blorg 1 2) => blorg(1, 2)

(foobar (blorg 1 2) (blabla 3 4) (array 2 3 4))

=> foobar(blorg(1, 2), blabla(3, 4), [ 2, 3, 4 ])

((aref foo i) 1 2) => foo[i](1, 2)

A method call is a function call where the function name is a symbol and begins
with a “.” . In a method call, the name of the function is append to its first
argument, thus reflecting the method call syntax of JavaScript. Please note that
most method calls can be abbreviated using the “.” trick in symbol names (see
“Symbol Conversion” above).

(.blorg this 1 2) => this.blorg(1, 2)

(this.blorg 1 2) => this.blorg(1, 2)

(.blorg (aref foobar 1) NIL T)

=> foobar[1].blorg(null, true)

16 Operator Expressions

; (operator {argument}*)

; (single-operator argument)

;

; operator ::= one of *, /, %, +, -, <<, >>, >>>, < >, EQL,

; ==, !=, =, ===, !==, &, ^, |, &&, AND, ||, OR.

; single-operator ::= one of INCF, DECF, ++, --, NOT, !

; argument ::= a ParenScript expression

Operator forms are similar to function call forms, but have an operator as func-
tion name.

Please note that = is converted to == in JavaScript. The = ParenScript op-
erator is not the assignment operator. Unlike JavaScript, ParenScript supports
multiple arguments to the operators.

(* 1 2) => 1 * 2

(= 1 2) => 1 == 2

(eql 1 2) => 1 == 2

Note that the resulting expression is correctly parenthesized, according to the
JavaScript operator precedence that can be found in table form at:

http://www.codehouse.com/javascript/precedence/

(* 1 (+ 2 3 4) 4 (/ 6 7))

=> 1 * (2 + 3 + 4) * 4 * (6 / 7)

The pre increment and decrement operators are also available. INCF and DECF
are the pre-incrementing and pre-decrementing operators. These operators can
take only one argument.

(incf i) => ++i

(decf i) => --i

The 1+ and 1- operators are shortforms for adding and substracting 1.

(1- i) => i - 1

(1+ i) => i + 1

The not operator actually optimizes the code a bit. If not is used on another
boolean-returning operator, the operator is reversed.

(not (< i 2)) => i >= 2

(not (eql i 2)) => i != 2

17 Body forms

; (PROGN {statement}*) in statement context

; (PROGN {expression}*) in expression context

;

; statement ::= a ParenScript statement

; expression ::= a ParenScript expression

The PROGN special form defines a sequence of statements when used in a state-
ment context, or sequence of expression when used in an expression context.
The PROGN special form is added implicitly around the branches of conditional
executions forms, function declarations and iteration constructs. For example,
in a statement context:

(progn (blorg i) (blafoo i))

=> blorg(i);

blafoo(i);

In an expression context:

(+ i (progn (blorg i) (blafoo i)))

=> i + (blorg(i), blafoo(i))

A PROGN form doesn’t lead to additional indentation or additional braces around
it’s body.

18 Function Definition
; (DEFUN name ({argument}*) body)

; (LAMBDA ({argument}*) body)

;

; name ::= a Lisp Symbol

; argument ::= a Lisp symbol

; body ::= a list of ParenScript statements

As in Lisp, functions are defined using the DEFUN form, which takes a name, a
list of arguments, and a function body. An implicit PROGN is added around the
body statements.

(defun a-function (a b)

(return (+ a b)))

=> function aFunction(a, b) {

return a + b;

}

Anonymous functions can be created using the LAMBDA form, which is the
same as DEFUN, but without function name. In fact, LAMBDA creates a DEFUN
with an empty function name.

(lambda (a b) (return (+ a b)))

=> function (a, b) {

return a + b;

}

19 Assignment

; (SETF {lhs rhs}*)

;

; lhs ::= a ParenScript left hand side expression

; rhs ::= a ParenScript expression

Assignment is done using the SETF form, which is transformed into a series of
assignments using the JavaScript = operator.

(setf a 1) => a = 1;

(setf a 2 b 3 c 4 x (+ a b c))

=> a = 2;

b = 3;

c = 4;

x = a + b + c;

The SETF form can transform assignments of a variable with an operator ex-
pression using this variable into a more “efficient” assignment operator form.
For example:

(setf a (1+ a)) => a++;

(setf a (+ a 2 3 4 a)) => a += 2 + 3 + 4 + a;

(setf a (- 1 a)) => a = 1 - a;

New types of setf places can be defined in one of two ways: using DEFSETF or
using DEFUN with a setf function name; both are analogous to their Common
Lisp counterparts. DEFSETF supports both long and short forms, while DEFUN
of a setf place generates a JavaScript function name with the setf prefix:

(defun (setf color) (new-color el)

(setf (slot-value (slot-value el ’style) ’color) new-color))

=> function __setf_color(newColor, el) {

el.style.color = newColor;

};

(setf (color some-div) (+ 23 "em"))

=> var _js2 = someDiv;

var _js1 = 23 + ’em’;

__setf_color(_js1, _js2);

Note that temporary variables are generated to preserve evaluation order of
the arguments as they would be in Lisp. The following example illustrates how
setf places can be used to provide a uniform protocol for positioning elements
in HTML pages:

(defsetf left (el) (offset)

‘(setf (slot-value (slot-value ,el ’style) ’left) ,offset)) => null

(setf (left some-div) (+ 123 "px"))

=> var _js2 = someDiv;

var _js1 = 123 + ’px’;

_js2.style.left = _js1;

(progn (defmacro left (el)

‘(slot-value ,el ’offset-left))

(left some-div))

=> someDiv.offsetLeft;

20 Single argument statements

; (RETURN {value}?)

; (THROW {value}?)

;

; value ::= a ParenScript expression

The single argument statements return and throw are generated by the form
RETURN and THROW. THROW has to be used inside a TRY form. RETURN is used
to return a value from a function call.

(return 1) => return 1

(throw "foobar") => throw ’foobar’

21 Single argument expression

; (DELETE {value})

; (VOID {value})

; (TYPEOF {value})

; (INSTANCEOF {value})

; (NEW {value})

;

; value ::= a ParenScript expression

The single argument expressions delete, void, typeof, instanceof and
new are generated by the forms DELETE, VOID, TYPEOF, INSTANCEOF and
NEW. They all take a ParenScript expression.

(delete (new (*foobar 2 3 4))) => delete new Foobar(2, 3, 4)

(if (= (typeof blorg) *string)

(alert (+ "blorg is a string: " blorg))

(alert "blorg is not a string"))

=> if (typeof blorg == String) {

alert(’blorg is a string: ’ + blorg);

} else {

alert(’blorg is not a string’);

}

22 Conditional Statements
; (IF conditional then {else})

; (WHEN condition then)

; (UNLESS condition then)

;

; condition ::= a ParenScript expression

; then ::= a ParenScript statement in statement context, a

; ParenScript expression in expression context

; else ::= a ParenScript statement in statement context, a

; ParenScript expression in expression context

The IF form compiles to the if javascript construct. An explicit PROGN around
the then branch and the else branch is needed if they consist of more than one
statement. When the IF form is used in an expression context, a JavaScript ?,
: operator form is generated.

(if (blorg.is-correct)

(progn (carry-on) (return i))

(alert "blorg is not correct!"))

=> if (blorg.isCorrect()) {

carryOn();

return i;

} else {

alert(’blorg is not correct!’);

}

(+ i (if (blorg.add-one) 1 2))

=> i + (blorg.addOne() ? 1 : 2)

The WHEN and UNLESS forms can be used as shortcuts for the IF form.

(when (blorg.is-correct)

(carry-on)

(return i))

=> if (blorg.isCorrect()) {

carryOn();

return i;

}

(unless (blorg.is-correct)

(alert "blorg is not correct!"))

=> if (!blorg.isCorrect()) {

alert(’blorg is not correct!’);

}

23 Variable declaration
; (DEFVAR var {value}?)

; (LET ({var | (var value)) body)

;

; var ::= a Lisp symbol

; value ::= a ParenScript expression

; body ::= a list of ParenScript statements

Variables (either local or global) can be declared using the DEFVAR form, which
is similar to its equivalent form in Lisp. The DEFVAR is converted to “var ... =
...” form in JavaScript.

(defvar *a* (array 1 2 3)) => var A = [ 1, 2, 3 ]

(if (= i 1)

(progn (defvar blorg "hallo")

(alert blorg))

(progn (defvar blorg "blitzel")

(alert blorg)))

=> if (i == 1) {

var blorg = ’hallo’;

alert(blorg);

} else {

var blorg = ’blitzel’;

alert(blorg);

}

Another way to declare local variables is to use the LET form. Note that the
ParenScript LET compiles to a straightforward assignment and does not have
lexical-scoping semantics, unlike its Lisp cousin.

(if (= i 1)

(let ((blorg "hallo"))

(alert blorg))

(let ((blorg "blitzel"))

(alert blorg)))

=> if (i == 1) {

var blorg = ’hallo’;

alert(blorg);

} else {

var blorg = ’blitzel’;

alert(blorg);

}

Moreover, beware that scoping in Lisp and JavaScript are quite different. For
example, don’t rely on closures capturing local variables in the way you’d
think they would.

24 Iteration constructs
; (DO ({var | (var {init}? {step}?)}*) (end-test) body)

; (DOTIMES (var numeric-form) body)

; (DOLIST (var list-form) body)

; (DOEACH (var object) body)

; (WHILE end-test body)

;

; var ::= a Lisp symbol

; numeric-form ::= a ParenScript expression resulting in a number

; list-form ::= a ParenScript expression resulting in an array

; object ::= a ParenScript expression resulting in an object

; init ::= a ParenScript expression

; step ::= a ParenScript expression

; end-test ::= a ParenScript expression

; body ::= a list of ParenScript statements

The DO form, which is similar to its Lisp form, is transformed into a JavaScript
for statement. Note that the ParenScript DO form does not have a return value,
that is because for is a statement and not an expression in JavaScript.

(do ((i 0 (1+ i))

(l (aref blorg i) (aref blorg i)))

((or (= i blorg.length)

(eql l "Fumitastic")))

(document.write (+ "L is " l)))

=> for (var i = 0, l = blorg[i];

!(i == blorg.length || l == ’Fumitastic’);

i = i + 1, l = blorg[i]) {

document.write(’L is ’ + l);

}

The DOTIMES form, which lets a variable iterate from 0 upto an end value, is a
shortcut for DO.

(dotimes (i blorg.length)

(document.write (+ "L is " (aref blorg i))))

=> for (var i = 0; i < blorg.length; i = i + 1) {

document.write(’L is ’ + blorg[i]);

}

The DOLIST form is a shortcut for iterating over an array. Note that this form
creates temporary variables using a function called PS-GENSYM, which is sim-
ilar to its Lisp counterpart GENSYM.

(dolist (l blorg)

(document.write (+ "L is " l)))

=> var tmpArr1 = blorg;

for (var tmpI2 = 0; tmpI2 < tmpArr1.length;

tmpI2 = tmpI2 + 1) {

var l = tmpArr1[tmpI2];

document.write(’L is ’ + l);

};

The DOEACH form is converted to a for (var .. in ..) form in JavaScript.
It is used to iterate over the enumerable properties of an object.

(doeach (i object)

(document.write (+ i " is " (aref object i))))

=> for (var i in object) {

document.write(i + ’ is ’ + object[i]);

}

The WHILE form is transformed to the JavaScript form while, and loops until
a termination test evaluates to false.

(while (film.is-not-finished)

(this.eat (new *popcorn)))

=> while (film.isNotFinished()) {

this.eat(new Popcorn);

}

25 The ‘CASE’ statement
; (CASE case-value clause*)

;

; clause ::= (value body) | ((value*) body) | t-clause

; case-value ::= a ParenScript expression

; value ::= a ParenScript expression

; t-clause ::= {t | otherwise | default} body

; body ::= a list of ParenScript statements

The Lisp CASE form is transformed to a switch statement in JavaScript. Note
that CASE is not an expression in ParenScript.

(case (aref blorg i)

((1 "one") (alert "one"))

(2 (alert "two"))

(t (alert "default clause")))

=> switch (blorg[i]) {

case 1:

case ’one’:

alert(’one’);

break;

case 2:

alert(’two’);

break;

default: alert(’default clause’);

}

; (SWITCH case-value clause*)

; clause ::= (value body) | (default body)

The SWITCH form is the equivalent to a javascript switch statement. No break
statements are inserted, and the default case is named DEFAULT. The CASE
form should be prefered in most cases.

(switch (aref blorg i)

(1 (alert "If I get here"))

(2 (alert "I also get here"))

(default (alert "I always get here")))

=> switch (blorg[i]) {

case 1: alert(’If I get here’);

case 2: alert(’I also get here’);

default: alert(’I always get here’);

}

26 The ‘WITH’ statement
; (WITH object body)

;

; object ::= a ParenScript expression evaluating to an object

; body ::= a list of ParenScript statements

The WITH form is compiled to a JavaScript with statements, and adds the ob-
ject object as an intermediary scope objects when executing the body.

(with (create :foo "foo" :i "i")

(alert (+ "i is now intermediary scoped: " i)))

=> with ({ foo : ’foo’,

i : ’i’ }) {

alert(’i is now intermediary scoped: ’ + i);

}

27 The ‘TRY’ statement
; (TRY body {(:CATCH (var) body)}? {(:FINALLY body)}?)

;

; body ::= a list of ParenScript statements

; var ::= a Lisp symbol

The TRY form is converted to a JavaScript try statement, and can be used to
catch expressions thrown by the THROW form. The body of the catch clause is
invoked when an exception is catched, and the body of the finally is always
invoked when leaving the body of the TRY form.

(try (throw "i")

(:catch (error)

(alert (+ "an error happened: " error)))

(:finally

(alert "Leaving the try form")))

=> try {

throw ’i’;

} catch (error) {

alert(’an error happened: ’ + error);

} finally {

alert(’Leaving the try form’);

}

28 The HTML Generator
; (PS-HTML html-expression)

The HTML generator of ParenScript is very similar to the htmlgen HTML gen-
erator library included with AllegroServe. It accepts the same input forms as
the AllegroServer HTML generator. However, non-HTML construct are com-
piled to JavaScript by the ParenScript compiler. The resulting expression is a
JavaScript expression.

(ps-html ((:a :href "foobar") "blorg"))

=> ’<a href=\"foobar\">blorg</a>’

(ps-html ((:a :href (generate-a-link)) "blorg"))

=> ’<a href=\"’ + generateALink() + ’\">blorg</a>’

We can recursively call the ParenScript compiler in an HTML expression.

(document.write

(ps-html ((:a :href "#"

:onclick (lisp (ps-inline (transport)))) "link")))

=> document.write(’<a href=\"#\" onclick=\"’ + ’javascript:transport()’ + ’\">link</a>’)

Forms may be used in attribute lists to conditionally generate the next attribute.
In this example the textarea is sometimes disabled.

(let ((disabled nil)

(authorized t))

(setf element.inner-h-t-m-l

(ps-html ((:textarea (or disabled (not authorized)) :disabled "disabled")

"Edit me"))))

=> var disabled = null;

var authorized = true;

element.innerHTML =

’<textarea’

+ (disabled || !authorized ? ’ disabled=\"’ + ’disabled’ + ’\"’ : ’’)

+ ’>Edit me</textarea>’;

; (CSS-INLINE css-expression)

Stylesheets can also be created in ParenScript.

(css-inline :color "red"

:font-size "x-small")

=> ’color:red;font-size:x-small’

(defun make-color-div(color-name)

(return (ps-html ((:div :style (css-inline :color color-name))

color-name " looks like this."))))

=> function makeColorDiv(colorName) {

return ’<div style=\"’ + (’color:’ + colorName) + ’\">’ + colorName

+ ’ looks like this.</div>’;

}

29 Macrology

; (DEFPSMACRO name lambda-list macro-body)

; (MACROLET ({name lambda-list macro-body}*) body)

; (SYMBOL-MACROLET ({name macro-body}*) body)

; (PS-GENSYM {string})

;

; name ::= a Lisp symbol

; lambda-list ::= a lambda list

; macro-body ::= a Lisp body evaluating to ParenScript code

; body ::= a list of ParenScript statements

; string ::= a string

ParenScript can be extended using macros, just like Lisp can be extended us-
ing Lisp macros. Using the special Lisp form DEFPSMACRO, the ParenScript
language can be extended. DEFPSMACRO adds the new macro to the toplevel
macro environment, which is always accessible during ParenScript compila-
tion. For example, the 1+ and 1- operators are implemented using macros.

(defpsmacro 1- (form)

‘(- ,form 1))

(defpsmacro 1+ (form)

‘(+ ,form 1))

A more complicated ParenScript macro example is the implementation of the
DOLIST form (note how PS-GENSYM, the ParenScript of GENSYM, is used to
generate new ParenScript variable names):

(defpsmacro dolist (i-array &rest body)

(let ((var (first i-array))

(array (second i-array))

(arrvar (ps-gensym "arr"))

(idx (ps-gensym "i")))

‘(let ((,arrvar ,array))

(do ((,idx 0 (incf ,idx)))

((>= ,idx (slot-value ,arrvar ’length)))

(let ((,var (aref ,arrvar ,idx)))

,@body)))))

Macros can be defined in ParenScript code itself (as opposed to from Lisp)
by using the ParenScript MACROLET and DEFMACRO forms. ParenScript also
supports the use of macros defined in the underlying Lisp environment. Ex-
isting Lisp macros can be imported into the ParenScript macro environment
by IMPORT-MACROS-FROM-LISP. This functionality enables code sharing be-
tween ParenScript and Lisp, and is useful in debugging since the full power
of Lisp macroexpanders, editors and other supporting facilities can be used.
However, it is important to note that the macroexpansion of Lisp macros and
ParenScript macros takes place in their own respective environments, and many
Lisp macros (especially those provided by the Lisp implementation) expand
into code that is not usable by ParenScript. To make it easy for users to take
advantage of these features, two additional macro definition facilities are pro-
vided by ParenScript: DEFMACRO/PS and DEFMACRO+PS. DEFMACRO/PS de-

fines a Lisp macro and then imports it into the ParenScript macro environ-
ment, while DEFMACRO+PS defines two macros with the same name and ex-
pansion, one in ParenScript and one in Lisp. DEFMACRO+PS is used when the
full ’macroexpand’ of the Lisp macro yields code that cannot be used by Paren-
Script. ParenScript also supports symbol macros, which can be introduced
using the ParenScript form SYMBOL-MACROLET.For example, the ParenScript
WITH-SLOTS is implemented using symbol macros.

(defjsmacro with-slots (slots object &rest body)

‘(symbol-macrolet ,(mapcar #’(lambda (slot)

‘(,slot ’(slot-value ,object ’,slot)))

slots)

,@body))

30 The ParenScript namespace system

; (setf (PS-PACKAGE-PREFIX package-designator) string)

Although JavaScript does not offer namespacing or a package system, Paren-
Script does provide a namespace mechanism for generated JavaScript by in-
tegrating with the Common Lisp package system. Since ParenScript code is
normally read in by the Lisp reader, all symbols (except for uninterned ones,
ie - those specified with the #: reader macro) have a Lisp package. By default,
no packages are prefixed. You can specify that symbols in a particular package
receive a prefix when translated to JavaScript with the PS-PACKAGE-PREFIX
place.

(defpackage "MY-LIBRARY"

(:use #:parenscript))

(setf (ps-package-prefix :my-library) "my_library_")

(defun my-library::library-function (x y)

(return (+ x y)))

-> function my_library_libraryFunction(x, y) {

return x + y;

}

31 Identifier obfuscation
; (OBFUSCATE-PACKAGE package-designator)

; (UNOBFUSCATE-PACKAGE package-designator)

Similar to the namespace mechanism, ParenScript provides a facility to gener-
ate obfuscated identifiers in certain Lisp packages.

(defpackage "OBFUSCATE-ME")

(obfuscate-package :obfuscate-me)

(defun obfuscate-me::library-function2 (a b obfuscate-me::foo)

(+ a (my-library::library-function b obfuscate-me::foo)))

The obfuscation and namespace facilities can be used on packages at the same
time.

32 The ParenScript Compiler

; (COMPILE-SCRIPT script-form &key (output-stream nil))

; (PS &body body)

; (PS* &body body)

; (PS-INLINE &body body)

; (LISP &body lisp-forms)

;

; body ::= ParenScript statements comprising an implicit ‘PROGN’

For static ParenScript code, the macros PS and PS-INLINE, avoid the need to
quote the ParenScript expression. PS* and COMPILE-SCRIPT evaluate their
arguments. All these forms except for COMPILE-SCRIPT treat the given forms
as an implicit PROGN. PS and PS* return a string of the compiled body, while
COMPILE-SCRIPT takes an optional output-stream parameter that can be used
to specify a stream to which the generated JavaScript will be written. PS-INLINE
generates a string that can be used in HTML node attributes. ParenScript can
also call out to arbitrary Lisp code at compile-time using the special form LISP.
This is typically used to insert the values of Lisp special variables into Paren-
Script code at compile-time, and can also be used to make nested calls to the
ParenScript compiler, which comes in useful when you want to use the result
of PS-INLINE in PS-HTML forms, for example. Alternatively the same thing
can be accomplished by constructing ParenScript programs as lists and passing
them to PS* or COMPILE-SCRIPT.

